REFERENCES

[1] G. A. Tyler and B. J. Thompson, "Fraunhofer holography applied to particle size analysis A reassessment," *Optica Acta*, vol. 23, no. 9, pp. 685-700, 1976.

[2] B. J. Thompson and P. Dunn, "Recent advences in holography," *Proc. of the Society of Photo-optical Instrumentation Engineers (SPIE)*, vol. 215, pp. 102-111, 1980.

[3] S. L. Cartwright, P. Dunn, and B. J. Thompson, "Particle sizing using far-field holography: new developements," *Optical Engineering*, vol. 19(5), pp. 727-733, Sept/Oct 1980.

[4] B. J. Thompson, "Holographic particle sizing techniques," *J. of Physics E*, pp 803-806, 1974.

[5] G. B. Parrent,Jr and B. J. Thompson, "On the Fraunhofer (far-field) diffraction patterns of opaque and transparent objects with coherent background," *Optica Acta*, vol. 11, pp. 183-192, 1964.

[6] B. J. Thompson, J. H. Ward, and W. R. Zinky, "Application of hologram techniques for particle size analysis," *Applied Optics*, vol. 6 no.3 pp. 519-526, March 1967.

[7] B. J. Thompson, "Holographic particle characterization - size, position and velocity," *Photomethods*, pp. 12-14, May 1984.

[8] E. A. Boettner and B. J. Thompson, "Multiple exposure holography of moving fibrous particulate matter in the respirotary range," *Optical Engineering*, vol. 12, no. 2, pp. 56-59, Mar-Apr 1973.

[9] R. Bexon, M. G. Dalzell, M. C. Stainer, "In-line holography and the assessment of aerosols," *Optics and Laser Technology*, pp. 161-165, 1976.

[10] J. D. Trolinger, "Particle field holography," *Optical Engineering*, vol. 14, no. 5, pp. 383-391, Sept-Oct 1975.

[11] K. L. Carder and D. J. Meyers, "Holography of settling particles: shape parameters," *Optical Engineering*, vol. 19, no. 5, pp. 735-738, Sept-Oct 1980.

[12] I. Prihyl and C. M. Vest, "Holographic imaging of semitransparent droplets or particles," *Applied Optics*, vol. 21, no. 14, pp. 2541-2547, 15 July 1982.

[13] R. Bexon, J. Gibbs, G. D. Bishop, "Automatic assessment of aerosol holograms," *J. Aerosol Sci.*, vol. 7, pp. 397-407, 1976.

[14] G. Haussmann and W. Lauterborn, " Determination of size position of fast moving gas bubbles in liquids by digital 3-D image processing of hologram reconstructions," *Applied Optics*, vol. 19, no. 20, pp. 3529-3535, 15 Oct 1980.

[15] D. Gabor, "A new microscopic principle," *Nature*, vol. 161, pp. 777-778, 1948.

[16] D. Gabor, "Microscopy by reconstructed wave-fronts," *Proc. Roy. Soc. (London)*, vol. A197, pp. 454-487, 1949.

[17] D. Gabor, "Microscopy by reconstructed wave-fronts, II," *Proc. Phys. Soc.*, vol. B64, pp. 449-469, 1951.

[18] E. N. Leith and J. Upatnieks, "Reconstructed wavefronts and communication theory," *Journal of the Optical Society of America*, vol. 52, no. 10, pp. 1123-1130, 1962.

[19] G. A. Tyler, "Image coding with far-field holograms," *Applied Optics*, vol. 17, no. 17, pp. 2768-2772, 1 Sep 1978.

[20] C. S. Vikram and L. Billet, "In-line Fraunhofer holography at a few far fields," *Applied Optics*, vol. 23, no. 18, pp. 3091-3094, 15 Sep 1984.

[21] K. H. S. Marie, A. P. Anderson, and J. C. Bennett, "Digital in-line holographic techniques for long wavelength imaging," *IEE Proceedings*, Pt H, vol. 129, no. 4, pp. 211-220, August 1982.

[22] K. H. S. Marie, J.C. Bennet, and A. P. Anderson, "Digital processing technique for suppressing the interfering outputs in the image from an in-line hologram," *Electronics Letters*, vol. 15, no. 8, pp. 241-243, 12 April 1979.

[23] K. H. S. Marie, J. C. Bennet, and A. P. Anderson, "Microwave images with reduced background effects from digitally processed in-line hoograms," *Electronics Letters*, vol. 16, no. 13, pp. 493-495, 19 June 1980.

[24] T. S. Huang, "Digital holography," *Proceedings of IEEE*, vol. 59, no. 9, pp. 1335-1346, September 1971.

[25] F. J. Paoloni and M. J. Duffy, "Digital image reconstruction of microwave holograms," *IEEE Trans. on Antennas and Propagation*, vol. AP-31, no. 2, pp. 389-394, March 1983.

[26] H. Shigesava, K. Takiyama, and M. Niskimura, "High quality image reconstruction from double microwave holograms," *IEEE Trans. Microwave Theory Techn.*, vol. MTT24, pp. 529-531, August 1976.

[27] Y. Aoki and A. Boivin "Computer reconstrcution of images from a microwave hologram," *Proc. IEEE*, vol. 58, pp. 821-822, May 1970.

[28] R. P. Porter and A. J. Devaney "Generalized holography and computational solution to inverse source problems," *J. Optical Society of America*, Vol. 72, no. 12

[29] W. H. Southwell, "Validity of the Fresnel approximation in the near field," *J. of the Optical Society of America*, vol. 71, no. 1, pp. 7-14, Jan 1981.

[30] R. H. Wade, "Spectral analysis of hologram and reconstructed images," *Optik*, vol. 40, no. 2, pp.201-216, 1974.

[31] M. Born and E. Wolf, *Principles of Optics*, Fifth Edition, Oxford, New York, Pergamon Press, 1975.

[32] L. M. Soroko, *Holography and Coherent Optics*, New York, Plenum Press, New York, 1980.

[33] A. Papoulis, *Systems and Transforms with Applications in Optics*, New York, McGraw-Hill, 1968.

[34] H. J, Caulfield, "Image degredation in undersampled holograms," *Physics Letters*, vol. 28A, no. 9, 10 Feb 1969.

[35] P. K. Murphy and N. C. Gallagher, "Hexagonal sampling techniques applied to Fourier and Fresnel digital holograms," *J. of the Optical Society of America*, vol. 72, no. 7, pp. 929-937, July 1982.

[36] W. T. Cathey,Jr. "The effect of finite sampling in holography," *Optik*, vol. 27, pp. 317-326, 1968.

[37] W. H. Carter, "Aliasing in sampled holograms," *Proc. IEEE (Letters)*, vol. 56, pp. 96-98, 1968.

[38] A. V. Oppenheim and R. W. Schafer, *Digital Signal Processing*, Englewood Cliffs, N.J., Prentice-Hall, 1975.

[39] D. E. Dudgeon and R. M. Mercereau, *Multidimensional Digital Signal Processing*, Englewood Cliffs, N.J., Prentice-Hall, 1984.

[40] L. R. Rabiner and B. Gold, *Theory and Application of Digital Signal Processing*, Englewood Cliffs, N.J., Prentice-Hall, 1975.

[41] A. Papoulis, *Signal Analysis*, New York, McGraw-Hill, 1977.

[42] R. E. Edwards, *Fourier Series a modern introduction*, Second edition, Volumes 1 & 2, New York, Springer Verlag, 1979.

[43] P. L. Butzer and R. J. Nessel, *Fourier Analysis and Approximation*, Volume 1, Basel, Birkhauser, 1971.

[44] C. Lanczos, *Discourse on Fourier Series*, New York, Hafner, 1966.

[45] A. Zygmund, *Trigonometric Series*, Volume 1, London, Cambridge University Press, 1968.

[46] E. C. Titchmarsh, *Introduction to the Theory of Fourier Integrals*, Second Edition, Oxford, The Clarendon Press, 1962.

[47] L. Kuipers and H. Niederreiter, *Uniform Distribution of Sequences*, New York, Wiley, 1974.

[48] R. K. Ritt, *Fourier Series*, New York, McGraw-Hill, 1970.

[49] T. Peli and D. Malah, "A study of edge detection algorithms," *Computer Graphics and Image Processing*, vol 20, pp.1-21, 1982.

[50] J. A. G. Halé, "Detection of elementary features in a picture by non-linear local numerical processing," *Proc. Third Int. Joint Conf. on Pattern Recognition*, pp. 764-768, 1976.

[51] W. K. Pratt, *Digital Image Processing*, New York, Wiley, 1977.

[52] J. S. Weszka, R. N. Nagel, A. Rosenfeld, "A threshold selection technique," *IEEE Trans. on Computers*, pp. 1322-1326, Dec 1974.

[53] L. S. Davis, "A survey of edge detection techniques," *Computer Graphics and Image Processing*, vol. 4, pp. 248-270, 1975.

[54] J. K. Aggarwal, R. O. Duda, and A. Rosenfeld - Editors, *Computer Methods in Image Analysis*, IEEE Press, New York, 1977.

[55] M. T. Heideman, D. H. Johnson, and C. S. Burrus, "Gauss and the history of the fast Fourier transform," *IEEE ASSP Magazine*, vol. 1, no. 4, pp. 14-21, October 1984.

[56] D. B. DeHaan, *Nouvelles Tables D'intégrales Définies*, New York, G. E. Stechert & Co., 1939.

[57] G. H. Liu, *A Digital Signal Processing Approach to Reconstruction from Holograms*, Ph. D. Dissertation, ECE Dept, SUNY at Buffalo, New York, Feb 1985.

[58] G. Liu, P. D. Scott, D. T. Shaw, "Modelling and sampling strategies for digital in-line holography" *Optical Society of America 1984 Annual Meeting*, October 1984.

[59] L. Onural, P. D. Scott, and D. T. Shaw "Digital algorithm extracting objects from their in-line holograms incorporating twin-image elimination," *Optical Society of America 1984 Annual Meeting*, October 1984.

[60] L. Onural and P. D. Scott, "Reconstruction of real 2-D signals from their in-line holograms by approximate inverse filtering," *Proc. Twenty-second Annual Allerton Conference on Communication, Control, and Computing*, October 1984.

[61] L. Onural and P.D. Scott, "A digital filtering system for decoding in-line holograms," *Proc. 1985 IEEE International Conference on Acoustics, Speech and Signal Processing, March 26-29 1985. (To appear.)*

APPENDIX A

COMPUTER PROGRAM FOR TYPE I SIMULATIONS

```
C************************************************************
C
C       THIS SUBROUTINE COMPUTES THE HOLOGRAM OF A 2-D COMPLEX
C       ARRAY OF DIMENSION N BY N.
C
C       INPUTS:
C               AR      :REAL PART OF INPUT ARRAY.
C               AI      :IMAGINARY PART OF INPUT ARRAY.
C               N1,M1   :DIMENSION OF THE ARRAYS (MUST BE SAME
C                        AS THE DIMENSION IN CALLING PROGRAM)
C                        (N1<513 AND M1<513 . FOR LARGER ARRAYS
C                         CHANGE THE DIMENSIONS OF FR AND FI)
C               N       :SIZE OF DATA (N BY N). (N<N1 AND N<M1)
C               AK      :THE NORMALIZED HOLOGRAM COEFFICIENT.
C                        (AK IS THE CONSTANT k IN THE HOLOGRAM
C                         FILTER FUNCTION exp[j(k**2)*pi/N*(x**2
C                                                      +y**2)]. )
c
C
C       OUTPUTS:
C               FR      :HOLOGRAM
C
C
C       THE INPUT IS ASSUMED TO BE THE LIGHT DISTRIBUTION AT
C       THE OBJECT PLANE. (OPAQUE=0)
C
C       AMPLITUDE AT THE OUTPUT IS ARBITRARY.
c
C
C**************************************************************
C
        SUBROUTINE HOLOS2D(AR,AI,N1,M1,N,AK)
        DIMENSION AR(N1,M1),AI(N1,M1)
C
C       HR AND HI ARE USED AS TEMPORARY STORAGE DURING
C       COMPUTATIONS.
C
        DIMENSION HR(512),HI(512)
C
C       FR AND FI ARE REQUIRED FOR SUBROUTINE FAS2.
C
        DIMENSION FR(512),FI(512)
        PARAMETER (PI=3.141593)
        CALL LIB$INIT_TIMER
C
```

```
C        TAKE FOURIER TRANSFORM OF INPUT HOLOGRAM.
C
         CALL FAS2(AR,AI,512,N1,M1,N,N,FR,FI)
         BK=1/AK
C
C        COMPUTE THE DFT OF HOLOGRAM FILTER COEFFICIENTS
C        AND DO THE FOURIER DOMAIN MULTIPLICATIONS,
C        USING SYMMETRY PROPERTIES TO REDUCE COMPUTATION TIME.
C        TEMPORARY STORAGE IS USED TO SIMPLIFY PROGRAMMING.
C
         DO 23 J=2,N/2
         DO 68 I=2,N/2
         SS=((I-1)*(I-1)+(J-1)*(J-1))*BK*BK/N*PI
         HR(I)=COS(SS)
         HR(N+2-I)=HR(I)
         HI(I)=SIN(SS)
         HI(N+2-I)=HI(I)
68       CONTINUE
         SS=((N/2)*(N/2)+(J-1)*(J-1))*BK*BK/N*PI
         HR(N/2+1)=COS(SS)
         HI(N/2+1)=SIN(SS)
         SS=((J-1)*(J-1))*BK*BK/N*PI
         HR(1)=COS(SS)
         HI(1)=SIN(SS)
         DO 95 I=1,N
         A=AR(I,J)
         AR(I,J)=AR(I,J)*HR(I)-AI(I,J)*HI(I)
         AI(I,J)=A*HI(I)+AI(I,J)*HR(I)
95       CONTINUE
         DO 96 I=1,N
         J1=N+2-J
         A=AR(I,J1)
         AR(I,J1)=AR(I,J1)*HR(I)-AI(I,J1)*HI(I)
         AI(I,J1)=A*HI(I)+AI(I,J1)*HR(I)
96       CONTINUE
23       CONTINUE
         DO 38 I=2,N/2
         SS=((I-1)*(I-1))*BK*BK/N*PI
         CSS=COS(SS)
         SSS=SIN(SS)
         A=AR(I,1)
         AR(I,1)=AR(I,1)*CSS-AI(I,1)*SSS
         AI(I,1)=A*SSS+AI(I,1)*CSS
         A=AR(N+2-I,1)
         AR(N+2-I,1)=AR(N+2-I,1)*CSS-AI(N+2-I,1)*SSS
         AI(N+2-I,1)=A*SSS+AI(N+2-I,1)*CSS
38       CONTINUE
         SS=((N/2)*(N/2))*BK*BK/N*PI
         CSS=COS(SS)
         SSS=SIN(SS)
         A=AR(N/2+1,1)
         AR(N/2+1,1)=AR(N/2+1,1)*CSS-AI(N/2+1,1)*SSS
         AI(N/2+1,1)=A*SSS+AI(N/2+1,1)*CSS
```

```
      DO 79 I=2,N/2
      SS=((I-1)*(I-1)+(N/2)*(N/2))*BK*BK/N*PI
      CSS=COS(SS)
      SSS=SIN(SS)
      J1=N/2+1
      A=AR(I,J1)
      AR(I,J1)=AR(I,J1)*CSS-AI(I,J1)*SSS
      AI(I,J1)=A*SSS+AI(I,J1)*CSS
      A=AR(N+2-I,J1)
      AR(N+2-I,J1)=AR(N+2-I,J1)*CSS-AI(N+2-I,J1)*SSS
      AI(N+2-I,J1)=A*SSS+AI(N+2-I,J1)*CSS
79    CONTINUE
      SS=((N/2)*(N/2)+(N/2)*(N/2))*BK*BK/N*PI
      CSS=COS(SS)
      SSS=SIN(SS)
      I=N/2+1
      A=AR(I,I)
      AR(I,I)=AR(I,I)*CSS-AI(I,I)*SSS
      AI(I,I)=A*SSS+AI(I,I)*CSS
C
C     TAKE THE INVERSE FOURIER TRANSFORM TO COMPLETE FILTERING
C
      CALL FAS2(AR,AI,512,N1,M1,-N,-N,FR,FI)
C
C     TAKE THE FINAL INTENSITY.
C
      DO 2 J=1,N
      DO 2 I=1,N
      AR(I,J)=AR(I,J)*AR(I,J)+AI(I,J)*AI(I,J)
2     CONTINUE
      CALL LIB$SHOW_TIMER
      RETURN
      END
```

APPENDIX B

COMPUTER PROGRAM FOR TYPE II SIMULATIONS

```
C*************************************************************
C
C       THIS SUBROUTINE COMPUTES THE HOLOGRAM OF A 2-D COMPLEX
C       ARRAY OF DIMENSION N BY N.
C
C       INPUTS:
C               AR       :REAL PART OF INPUT ARRAY.
C               AI       :IMAGINARY PART OF INPUT ARRAY.
C               N1,M1    :DIMENSION OF THE ARRAYS (MUST BE SAME
C                         AS THE DIMENSION IN CALLING PROGRAM)
C                         (N1<513 AND M1<513 . FOR LARGER ARRAYS
C                          CHANGE THE DIMENSIONS OF FR AND FI)
C               N        :SIZE OF DATA (N BY N). (N<N1 AND N<M1)
C               IL       :ILLUMINATION INDICATOR:
C                                 0 .... NO ILLUMINATION
C                                 1 .... ILLUMINATION
C                                        (INPUT IS REPLACED
C                                         BY 255-INPUT IF IL=1)
C               SRNI     :NORMALIZED SAMPLING PERIOD AT INPUT.
C
C       OUTPUTS:
C               FR       :HOLOGRAM
C               SRNO     :NORMALIZED SAMPLING PERIOD AT OUTPUT.
C
C
C*************************************************************
C
        SUBROUTINE HOLOF2D(AR,AI,N1,M1,N,SRNI,SRNO,IL)
        DIMENSION AR(N1,M1),AI(N1,M1)
        DIMENSION FR(512),FI(512)
        PARAMETER (PI=3.1415927)
        SRNO=1/SRNI
C
C       START THE MULTIPLICATIONS.
C
        DO 1 I=0,N-1
        DO 1 J=0,I
        AA=(I*I+J*J)*PI*SRNI*SRNI/FLOAT(N)
        A1=COS(AA)
        A2=SIN(AA)
        AX=AR(I+1,J+1)
        AR(I+1,J+1)=AX*A1-AI(I+1,J+1)*A2
        AI(I+1,J+1)=AX*A2+AI(I+1,J+1)*A1
        IF(I.EQ.J)GO TO 1
        AX=AR(J+1,I+1)
        AR(J+1,I+1)=AX*A1-AI(J+1,I+1)*A2
```

```
      AI(J+1,I+1)=AX*A2+AI(J+1,I+1)*A1
1     CONTINUE
C
C
C     TAKE THE DFT.
C
      CALL FAS2(AR,AI,512,N1,M1,N,N,FR,FI)
C
C     TAKE THE FINAL INTENSITY.
C
      DO 2 J=1,N
      DO 2 I=1,N
      AR(I,J)=AR(I,J)*AR(I,J)+AI(I,J)*AI(I,J)
2     CONTINUE
      RETURN
      END
```

APPENDIX C

COMPUTER PROGRAM FOR RECONSTRUCTION WITH TWIN IMAGE SUPPRESSION

```
C***********************************************************
C
C       THIS SUBROUTINE PERFORMS 2D HOLOGRAM RECONSTRUCTION
C       WITH TWIN-IMAGE SUPPRESSION.
C
C       INPUTS  :
C               FR      : 2D INPUT HOLOGRAM ARRAY.
C               FI      : ANOTHER 2D ARRAY WITH SAME SIZE AS FR.
C               HR      : A 1D ARRAY.
C               HI      : A 1D ARRAY.
C               MD      : DIMENSION OF ARRAYS; FR(MD,MD),
C                         FI(MD,MD) AND HR(MD),HI(MD).
C                         THESE ARRAYS MUST HAVE THE SAME
C                         DIMENSION IN THE CALLING PROGRAM.
C
C                         (FI,HR, AND HI ARE USED FOR
C                          INTERMEDIATE COMPUTATIONS.)
C
C               N       : SIZE OF DATA (N BY N) IN FR.
C               AK      : NORMALIZED HOLOGRAM COEFFICIENT.
C                         ( THIS COEFFICIENT IS a IN THE
C                           HOLOGRAM FILTER
c                           exp[ j a**2 pi/N (n**2 + m**2)].)
C
C               M       : NUMBER OF STAGES OF APPROXIMATE
C                         INVERSE FILTER.
C
C       OUTPUT:
C               FR      : RECONSTRUCTED IMAGE.
C
C*********************************************************
C
C
        SUBROUTINE HTT2D(FR,FI,MD,N,AK,M,HR,HI)
        DIMENSION FR(MD,MD),FI(MD,MD)
        DIMENSION HR(MD),HI(MD)
        DOUBLE PRECISION DC
        PARAMETER (PI=3.141593)
C       DC LEVEL IS ELIMINATED TO PREVENT STRONG EDGE EFFECTS
        DC=0.0
        DO 108 J=1,N
        DO 108 I=1,N
        DC=DC+FR(I,J)
108     CONTINUE
        DCS=DC/(N*N)
```

```
          DO 109 J=1,N
          DO 109 I=1,N
          FR(I,J)=FR(I,J)-DCS
109       CONTINUE
C         DOUBLE THE SIZE FOR DFT DOMAIN COMPUTATIONS
          N=2*N
          CALL FAS2(FR,FI,MD,MD,MD,N,N,HR,HI)
C         DC LEVEL IS NOT IMPORTANT, SO IT IS ELIMINATED.
          FR(1,1)=0
          FI(1,1)=0
C         ADJUST THE HOLOGRAM PARAMETER TO THE NEW SIZE.
          AK=AK*SQRT(2.0)
          BK=1/AK
C
C         START FILTERING BY USING SYMMETRY PROPERTIES
C         TO ELIMINATE REDUNDANT COMPUTATIONS
C
          DO 23 J=2,N/2
          DO 23 I=J+1,N/2
          CALL APRINV(I,J,BK,N,M,CC)
          FR(I,J)=FR(I,J)*CC
          FR(N+2-I,J)=FR(N+2-I,J)*CC
          FR(J,I)=FR(J,I)*CC
          FR(J,N+2-I)=FR(J,N+2-I)*CC
C
          FI(I,J)=FI(I,J)*CC
          FI(N+2-I,J)=FI(N+2-I,J)*CC
          FI(J,I)=FI(J,I)*CC
          FI(J,N+2-I)=FI(J,N+2-I)*CC
C
23        CONTINUE
C
C
          DO 24 I=2,N/2
          J=I
          CALL APRINV(I,J,BK,N,M,CC)
          FR(I,J)=FR(I,J)*CC
          FR(N+2-I,J)=FR(N+2-I,J)*CC
C
          FI(I,J)=FI(I,J)*CC
          FI(N+2-I,J)=FI(N+2-I,J)*CC
C
24        CONTINUE
C
          DO 25 I=2,N/2
          J=N/2+1
          CALL APRINV(I,J,BK,N,M,CC)
          FR(I,J)=FR(I,J)*CC
          FR(J,I)=FR(J,I)*CC
C
          FI(I,J)=FI(I,J)*CC
          FI(J,I)=FI(J,I)*CC
C
25        CONTINUE
```

```
C
         DO 26 I=2,N/2
         J=1
         CALL APRINV(I,J,BK,N,M,CC)
         FR(I,J)=FR(I,J)*CC
         FR(J,I)=FR(J,I)*CC
C
         FI(I,J)=FI(I,J)*CC
         FI(J,I)=FI(J,I)*CC
C
26       CONTINUE
         I=N/2+1
         J=N/2+1
         CALL APRINV(I,J,BK,N,M,CC)
         FR(I,J)=FR(I,J)*CC
C
         J=1
         CALL APRINV(I,J,BK,N,M,CC)
         FR(I,J)=FR(I,J)*CC
C
         FR(J,I)=FR(J,I)*CC
C
C        GET THE FULL ARRAY USING SYMMETRY PROPERTIES
         DO 21 J=2,N-1
         DO 21 I=2,N+1-J
         FR(N+2-I,N+2-J)=FR(I,J)
         FI(N+2-I,N+2-J)=-FI(I,J)
21       CONTINUE
         DO 22 I=2,N/2
         FR(I,N+2-I)=FR(N+2-I,I)
         FR(1,N+2-I)=FR(1,I)
         FR(N+2-I,1)=FR(I,1)
         FI(I,N+2-I)=-FI(N+2-I,I)
         FI(1,N+2-I)=-FI(1,I)
         FI(N+2-I,1)=-FI(I,1)
22       CONTINUE
C        TAKE INVERSE DFT TO COMPLETE FILTERING
         CALL FAS2(FR,FI,MD,MD,MD,-N,-N,HR,HI)
         N=N/2
         RETURN
         END
C
C
C***********************************************************
C
C        THIS SUBROUTINE COMPUTES THE APPROXIMATE INVERSE
C        FILTER ELEMENTS.
C
C        INPUTS:
C                I,J      :LOCATION.
C                BK       :HOLOGRAM COEFFICIENT.
C                          ( THIS IS THE PARAMETER b IN THE
C                            FOURIER TRANSFORM OF THE
C                            HOLOGRAM FILTER,
c                            HOLOGRAM FILTER,
C                            exp [-j 1/b**2 pi/N (k**2 + 1**2)].)
```

```
C
C                 N         :DATA SIZE (N BY N).
C                 M         :NUMBER OF STAGES IN THE APPROXIMATE
C                            INVERSE FILTER.
C
C       OUTPUT:
C                 CC        :FILTER ELEMENT IN FOURIER TRANSFORM
C                            DOMAIN.
C
C**************************************************************
C
C
        SUBROUTINE APRINV(I,J,BK,N,M,CC)
        PARAMETER (PI=3.141593)
        Z=((I-1)*(I-1) + (J-1)*(J-1))*BK*BK*PI/N
        CC=COS(Z)
        IF(M.GT.1)THEN
        SS=2*SIN(Z)
        CC=CC+SS*SIN(2*Z)
        SS=SS*2*COS(2*Z)
        ENDIF
        DO 11 K=3,M
        CC=CC+SS*SIN(2**(K-1)*Z)
        SS=SS*2*COS(2**(K-1)*Z)
11      CONTINUE
        CC=2*CC/M
        RETURN
        END
```

APPENDIX D

COMPUTER PROGRAM FOR COVARIANCE FILTER

```
C************************************************************
C
C       THIS SUBROUTINE IMPLEMENTS THE COVARIANCE FILTER.
C
C       INPUTS  :
C               IX      : 2-D INPUT INTEGER ARRAY OF DIMENSION
C                         NN,MM.
C
C               NN,MM   : DIMENSIONS OF THE ARRAYS.
C
C               NS,MS   : SIZE OF 2-D DATA IN THE ARRAYS.
C
C               MW      : WINDOW SIZE.
C
C       OUTPUT  :
C               AR      : 2-D OUTPUT ARRAY (REAL*4 TYPE)
C                         WITH DIMENSION NN,MM.
C
C***********************************************************
C
        SUBROUTINE COVAR(IX,AR,NN,MM,NS,MS,MW)
        INTEGER IX(NN,MM)
        DIMENSION AR(NN,MM)
C       TEST IF WINDOW SIZE IS ODD.
        IF(MW/2*2.NE.MW)GO TO 1
        TYPE *,'WINDOW SIZE MUST BE ODD'
        RETURN
1       CONTINUE
        N2=MW/2
C       START COVARIANCE FILTERING
        DO 2 L=N2+1,NS-N2
        DO 2 K=N2+1,MS-N2
C       AV IS THE LOCAL AVERAGE
        IAV=0
        DO 3 N=-N2,N2
        DO 3 M=-N2,N2
        IAV=IAV+IX(K-M,L-N)
3       CONTINUE
        AV=IAV/FLOAT(MW*MW)
C       S2 IS THE LOCAL COVARIANCE
        S2=0.0
        DO 4 N=-N2,N2
        DO 4 M=-N2,N2
        S2=S2+(IX(K-M,L-N)-AV) ** 2
4       CONTINUE
        S2=S2/(MW*MW)
        AR(K,L)=S2
2       CONTINUE
        RETURN
        END
```

APPENDIX E

UNIFORM DISTRIBUTION OF SEQUENCES

The uniform distribution of a sequence modulo 1 is defined in Ref.47, page 1. The definition and related mathematics provides a proof for the convergence of the average (as given in the text) of partial sums of Fourier series expansion, as follows:

The sequence $(x_n): p2^n$, where p is irrational, is an irrational sequence, since all x_n's are irrational. Therefore,

$$\Delta x_n = x_{n+1} - x_n = p\,(2^{n+1} - 2^n) = p\,2^n \ . \tag{E.1}$$

Therefore Δx_n is irrational for all n, which readily implies that it is also irrational as $n \to \infty$. So, (x_n) is uniformly distributed modulo 1 in the interval [0,1). (See Ref.47, page 28 theorem 3.3.) Since Δx_n is irrational as $n \to \infty$,

$$\lim_{N \to \infty} \frac{1}{N} \sum_{n=1}^{N} \cos(2\pi p 2^n) = \int_0^1 \cos 2\pi x \, dx = \left. \frac{\sin 2\pi x}{2\pi} \right|_0^1 = 0 \ . \tag{E.2}$$

(See Ref.47 page 2, theorem 1.1.)

Note that 2^n can be replaced by any other rational function and still the convergence to 0 is attained.

APPENDIX F

IMAGE PROCESSING ENVIRONEMENT AND COMPUTATIONAL ASPECTS

The implementation of the image processing algorithms given in this dissertation are implemented in an image processing system which consists of two dedicated mini-computers and a special purpose image processor. The computers are a VAX 11/730 and a VAX 11/750. Also, two other computers can be reached (a VAX 11/785 and a VAX 11/780) if needed. All these computers are virtual memory machines, which is very convenient for image processing. The bank of computers is linked to a COMTAL Vision One/20 image processor through the VAX 11/730. (See the figure below.) The image processor is capable of implementing many simple and standard image processing algorithms in a very short time as a consequence of its highly special parallel structured hardware. The processor is able to display a 512 by 512 picture consisting of 8-bit (24 if color image) pixels. (256 gray levels or 256^3 color combinations.) The processor is able to receive input from either the computers, or from a TV camera. If the input is analog, as in the TV camera, the real time digitizer of the processor handles the analog-to-digital conversion. Images can be displayed at the system monitor or can be photographed by a hard copy unit (a MATRIX INSTRUMENTS color graphic recorder). The images are transferred to the computer for specialized processing. Resultant images are sent back to the processor for display.

The holograms are digitized by using enlargments and the system digitizer as explained above. At the present time, the physical holography set-up is not directly coupled with the image processing system. The digitized hologram sections are sometimes preprocessed by the image processor, for some simple tasks as changing the DC level. Then they are transferred to a computer, where the subroutines given in previous appendices are implemented. The typical computation time for any one of the hologram reconstruction algorithms is in the range of 4-6 minutes on 785, with a floating point accelerator and using 1Mbyte physical memeory. The computation times on 750, without a floating point accelerator, and with same amount of physical memeory is about 12-18 minutes. These timings include the necessary processing performed by the main programs before and after the calling the subroutines given.



Figure A1. The image processing system.

APPENDIX G

THE OPTICAL HOLOGRAPHY SET-UP

The optical holography set-up which is used to for optical holograms of this dissertation is shown in the figure below. The laser used is a 35 mw helium-neon laser operating at 632.8 nm wavelength. Stationary particles are used as objects which were deposited on a glass. Most of the time, the particles were aliminum oxide with typical size range of 10-30 μm.



APPENDIX H

SOME USEFUL FORMULAS

$$\mathcal{F}_{2D}\left\{\exp\left[ja(x^2+y^2)\right]\right\} = j\frac{\pi}{a}\exp\left[-j\frac{u^2+v^2}{4a}\right] \tag{H.1}$$

$$\mathcal{F}_{2D}\left\{\cos a(x^2+y^2)\right\} = \frac{\pi}{a}\sin\frac{u^2+v^2}{4a} \tag{H.2}$$

$$\mathcal{F}_{2D}\left\{\sin a(x^2+y^2)\right\} = \frac{\pi}{a}\cos\frac{u^2+v^2}{4a} \tag{H.3}$$

$$\cos a(x^2+y^2) ** \cos a(x^2+y^2) = \frac{1}{2}\left[\frac{\pi}{a}\right]^2\delta(x,y) - \frac{1}{4}\frac{\pi}{a}\sin\frac{a}{2}(x^2+y^2) \tag{H.4}$$

$$\sin a(x^2+y^2) ** \sin a(x^2+y^2) = \frac{1}{2}\left[\frac{\pi}{a}\right]^2\delta(x,y) + \frac{1}{4}\frac{\pi}{a}\sin\frac{a}{2}(x^2+y^2) \tag{H.5}$$

$$\cos a(x^2+y^2) ** \sin a(x^2+y^2) = \frac{1}{4}\frac{\pi}{a}\cos\frac{a}{2}(x^2+y^2) \tag{H.6}$$